

```
1  ' -----
2  '       From the book Inside Active Directory, ISBN 0-201-61621-1
3  '       Copyright (C) 2002 by Addison-Wesley
4  '       Script by Sakari Kouti (see http://www.kouti.com)
5  ' You have a royalty-free right to use, modify, reproduce and distribute
6  ' this script (and/or any modified version) in any way you find useful,
7  ' provided that you agree that Addison-Wesley or Sakari Kouti has no
8  ' warranty, obligations or liability for the script. If you modify
9  ' the script, you must retain this copyright notice.
10 ' -----
11 Option Explicit
12
13 Dim objExcel, objDSE, objSchema, objChild, i, varValue
14 Dim arrParamList(14)
15
16 'Names and Identifiers
17 arrParamList(0) = "LDAPDisplayName"
18 arrParamList(1) = "cn"
19 arrParamList(2) = "adminDisplayName"
20 arrParamList(3) = "distinguishedName"
21 arrParamList(4) = "adminDescription"
22 arrParamList(5) = "governsID"
23 arrParamList(6) = "rDNAttID"
24 'Structure and Containment Rules: None
25 'Class Inheritance
26 arrParamList(7) = "objectClassCategory"
27 arrParamList(8) = "subClassOf"
28 'Miscellaneous
29 arrParamList(9) = "defaultObjectCategory"
30 arrParamList(10) = "defaultHidingValue"
31 arrParamList(11) = "defaultSecurityDescriptor"
32 arrParamList(12) = "systemOnly"
33 arrParamList(13) = "systemFlags"
34 arrParamList(14) = "objectCategory"
35
36 Set objExcel = WScript.CreateObject("Excel.Application")
37 objExcel.Visible = True
38 objExcel.Workbooks.Add
39 objExcel.ActiveSheet.Name = "classSchema objects"
40
41 objExcel.ActiveSheet.Range("A1").Activate
42 For i = LBound(arrParamList) To UBound(arrParamList)
43     objExcel.ActiveCell.Offset(0,i).Value = arrParamList(i)
44 Next
45 objExcel.ActiveCell.Offset(1,0).Activate    'Move to next line
46
47 Set objDSE = GetObject("LDAP://rootDSE")
48 Set objSchema = GetObject("LDAP://" & _
49     objDSE.Get("schemaNamingContext"))
50
51 On Error Resume Next
52 For Each objChild In objSchema
53     If objChild.Class = "classSchema" Then
54         For i = LBound(arrParamList) To UBound(arrParamList)
55             varValue = objChild.Get(arrParamList(i))
56             If Err <> 0 Then    'property not found or some other error
57                 Err.Clear
58                 varValue = ""
59             End If
60             objExcel.ActiveCell.Offset(0,i).Value = varValue
61         Next
62         objExcel.ActiveCell.Offset(1,0).Activate    'Move to next line
63     End If
64 Next
65
```